// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/TaskChainPresale.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/security/Pausable.sol";

interface Aggregator {
    function latestRoundData()
        external
        view
        returns (
            uint80 roundId,
            int256 answer,
            uint256 startedAt,
            uint256 updatedAt,
            uint80 answeredInRound
        );
}

contract TaskChainPresale is Ownable, ReentrancyGuard, Pausable {
    // Addresses for the various roles in the contract
    address private fundsAddress;
    address private tokenHolderAddress;
    address public saleTokenAddress =
        0x4C2e29dbc437C4b781963E5B2B393b1D4ea64b19;

    // Tokens and prices per stage
    uint256[11] public tokensInStages;
    uint256[11] public pricePerStage;

    // The token balance for each buyer
    mapping(address => uint256) public tokensBought;

    // Track which buyers have claimed their tokens
    mapping(address => bool) public tokensClaimed;
    mapping(uint256 => uint256) public tokensSoldPerStage;

    // Interfaces for the USDT token and the oracle price feed
    IERC20 public USDTInterface;
    Aggregator public priceFeed;

    // Track whether claims are active and the current sale stage
    bool public claimActive;
    Stages public stage;

    // Stages of the sale
    enum Stages {
        BETA_STAGE,
        STAGE_1,
        STAGE_2,
        STAGE_3,
        STAGE_4,
        STAGE_5,
        STAGE_6,
        STAGE_7,
        STAGE_8,
        STAGE_9,
        STAGE_10
    }

    // Events to be emitted by the contract
    event claimStart(bool claimActive);
    event TokenBought(address indexed buyer, uint256 amount, uint256 timestamp);
    event TokensClaimed(
        address indexed claimer,
        uint256 amount,
        uint256 timestamp
    );

    /**
     * @dev Constructor for the presale contract
     * @param _usdt address to send USDT token
     * @param _oracle Oracle contract to fetch ETH/USDT price
     */

    constructor(
        address _usdt,
        address _oracle
    ) ReentrancyGuard() nonZeroAddress(_usdt) nonZeroAddress(_oracle) {
        USDTInterface = IERC20(_usdt);
        priceFeed = Aggregator(_oracle);

        tokenHolderAddress = 0xc493fD92b08579682935520e9779Ee71CbD29A09;
        fundsAddress = 0xC35A5df1Bc74C7e96AfA6e047890f46459369743; 
        claimActive = false;
        stage = Stages.BETA_STAGE;
        tokensInStages = [
            280_000_000,
            300_000_000,
            300_000_000,
            200_000_000,
            200_000_000,
            200_000_000,
            200_000_000,
            200_000_000,
            200_000_000,
            360_000_000,
            360_000_000
        ];
        pricePerStage = [
            4000_000_000_000_000,
            5000_000_000_000_000,
            5200_000_000_000_000,
            5500_000_000_000_000,
            5700_000_000_000_000,
            6000_000_000_000_000,
            6500_000_000_000_000,
            8000_000_000_000_000,
            8500_000_000_000_000,
            9000_000_000_000_000,
            9500_000_000_000_000
        ];
    }

    modifier claimable() {
        require(claimActive == true, "Claiming is not active yet.");
        _;
    }

    modifier nonZeroAddress(address _addr) {
        require(_addr != address(0), "Address can't be zero");
        _;
    }

    /**
     * @dev buyTokenWithETH allows users to buy tokens with ETH
     * @param _tokenAmount total number of tokens user wants to buy
     */

    function buyTokenWithETH(
        uint256 _tokenAmount
    ) external payable nonReentrant whenNotPaused {
        require(_tokenAmount > 0, "Cannot buy zero tokens.");
        require(
            _tokenAmount <= getAvailableTokensInStage(),
            "Insufficient tokens available."
        );
        uint256 ethCost = calculateEthCost(_tokenAmount);
        require(msg.value >= ethCost, "Not enough ETH sent.");
        // Calculate excess ETH by user
        uint256 excessAmount = msg.value - ethCost;

        // Refund the excess ETH to the buyer
        if (excessAmount > 0) {
            (bool successTransferExcess, ) = payable(msg.sender).call{
                value: excessAmount
            }("");
            require(
                successTransferExcess,
                "Failed to send ETH to funds address."
            );
        }

        tokensSoldPerStage[uint256(stage)] += _tokenAmount;
        tokensBought[msg.sender] += _tokenAmount;

        (bool success, ) = payable(fundsAddress).call{value: ethCost}("");
        require(success, "Failed to send ETH to funds address.");

        emit TokenBought(msg.sender, _tokenAmount, block.timestamp);
    }

    /**
     * @dev buyTokenWithUSDT allows users to buy tokens with USDT
     * @param _tokenAmount total number of tokens user wants to buy
     */

    function buyTokenWithUSDT(
        uint256 _tokenAmount
    ) public nonReentrant whenNotPaused {
        require(_tokenAmount > 0, "Cannot buy zero tokens.");
        require(
            _tokenAmount <= getAvailableTokensInStage(),
            "Insufficient tokens available."
        );
        uint256 usdtCost = calculateUsdtCost(_tokenAmount);
        tokensSoldPerStage[uint256(stage)] += _tokenAmount;
        tokensBought[msg.sender] += _tokenAmount;
        require(
            USDTInterface.allowance(msg.sender, address(this)) >= usdtCost,
            "Not enough allowance"
        );

        require(
            USDTInterface.balanceOf(msg.sender) >= usdtCost,
            "Not enough balance"
        );
        USDTInterface.transferFrom(msg.sender, fundsAddress, usdtCost);
        emit TokenBought(msg.sender, _tokenAmount, block.timestamp);
    }

    /**
     * @dev claimTokens allows users to claim tokens after the sale is over
     */

    function claimTokens() external claimable nonReentrant {
        require(
            tokensBought[msg.sender] > 0,
            "You have not bought any tokens."
        );
        require(
            !tokensClaimed[msg.sender],
            "You have already claimed your tokens."
        );

        uint256 amountToClaim = tokensBought[msg.sender];
        require(amountToClaim > 0, "No tokens to claim.");
        tokensClaimed[msg.sender] = true;
        IERC20(saleTokenAddress).transferFrom(
            tokenHolderAddress,
            msg.sender,
            amountToClaim * 10 ** 18
        );
        emit TokensClaimed(msg.sender, amountToClaim, block.timestamp);
    }

    /**
     * @dev calculateEthCost helper function to calculate ETH cost for a given token amount
     */

    function calculateEthCost(
        uint256 _tokenAmount
    ) public view returns (uint256) {
        uint256 price = pricePerStage[uint256(stage)];
        (, int256 priceETH, , , ) = priceFeed.latestRoundData();
        uint256 ethCost = (_tokenAmount * price * 10 ** 18) /
            (uint256(priceETH) * 10 ** 10);
        return ethCost;
    }

    /**
     * @dev EthToTokenHelper function to calculate ETH cost for a given token amount
     */
    function EthToTokenHelper(uint _ethAmount) public view returns (uint256) {
        uint256 price = pricePerStage[uint256(stage)];
        (, int256 priceETH, , , ) = priceFeed.latestRoundData();
        uint256 tokenAmount = (_ethAmount * uint256(priceETH) * 10 ** 10) /
            (price * 10 ** 18);
        return tokenAmount;
    }

    /**
     * @dev USDTToTokenHelper function to calculate ETH cost for a given token amount
     */
    function USDTToTokenHelper(uint _usdtAmount) public view returns (uint256) {
        uint256 price = pricePerStage[uint256(stage)];
        uint256 tokenAmount = (_usdtAmount * 10 ** 12) / price;
        return tokenAmount;
    }

    /**
     * @dev calculateUsdtCost helper function to calculate UST cost for a given token amount
     */

    function calculateUsdtCost(
        uint256 _tokenAmount
    ) public view returns (uint256) {
        uint256 price = pricePerStage[uint256(stage)];
        uint256 usdtCost = (_tokenAmount * price) / (10 ** 12);
        return usdtCost;
    }

    function getAvailableTokensInStage() public view returns (uint256) {
        return
            tokensInStages[uint256(stage)] - tokensSoldPerStage[uint256(stage)];
    }

    /**
     * @dev setFundsAddress allows owner to set the funds address
     * @param _fundsAddress address of the funds address
     */
    function setFundsAddress(
        address _fundsAddress
    ) external onlyOwner nonZeroAddress(fundsAddress) {
        fundsAddress = _fundsAddress;
    }

    /**
     * @dev setTokenHolder allows owner to set the token holder address
     * @param _tokenHolder address of the token holder
     */

    function setTokenHolder(
        address _tokenHolder
    ) external onlyOwner nonZeroAddress(_tokenHolder) {
        tokenHolderAddress = _tokenHolder;
    }


    /**
     * @dev switchStage allows owner to switch to the next stage
     */

    function switchStage() external onlyOwner {
        require(
            uint256(stage) >= 0 && uint256(stage) <= 10,
            "Invalid stage transition."
        );

        // If it's not the first stage, move the remaining tokens to the next stage
        if (uint256(stage) >= 0 && uint256(stage) <= 9) {
            uint256 remainingTokens = tokensInStages[uint256(stage)] -
                tokensSoldPerStage[uint256(stage)];

            tokensInStages[uint256(stage) + 1] += remainingTokens;
        }

        // Move to the next stage
        stage = Stages(uint256(stage) + 1);
    }

    /**
     * @dev startClaim allows owner to start the claiming process
     */

    function startClaim() external onlyOwner {
        claimActive = true;
        emit claimStart(claimActive);
    }

    /**
     * @dev pauseContract allows owner to pause the contract
     */

    function pauseContract() external onlyOwner {
        _pause();
    }

    /**
     * @dev unpauseContract allows owner to unpause the contract
     */

    function unpauseContract() external onlyOwner {
        _unpause();
    }
}